# Maintainer: Johannes Löthberg <johannes@kyriasis.com>
# Contributor: Aaron DeVore <aaron.devore@gmail.com>

pkgname=python-parse
pkgver=1.19.0
pkgrel=4

pkgdesc='Parse strings using a specification based on the Python format() syntax.'
url='https://pypi.python.org/pypi/parse'
arch=('any')
license=('BSD')

depends=('python')
makedepends=('python-setuptools')

source=("https://pypi.org/packages/source/p/parse/parse-$pkgver.tar.gz")

sha256sums=('9ff82852bcb65d139813e2a5197627a94966245c897796760a3a2a8eb66f020b')

build() {
  cd "$srcdir"/parse-$pkgver
  python setup.py build
}

check() {
  cd "$srcdir"/parse-$pkgver
  python test_parse.py
}


package() {
  cd "$srcdir"/parse-$pkgver
  python setup.py install --root="$pkgdir" --optimize=1
  install -Dm644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}
